


push: Multiple Register Save (str to stack)


save registers
push{r4-r6, r8, fp, lr}


Registers are pushed on to the stack in order
right (high memory) to left (low memory)



- push copies the contents of the {reg list} to stack segment memory
- push Also subtracts (# of registers saved) * (4 bytes) from the sp to allocate space on the stack
 - sp = sp (# registers saved * 4)
- this must always be true: sp % 8 == 0

pop: Multiple Register Restore (ldr from stack)






stack segment low memory


- pop copies the contents of stack segment memory to the {reg list}
- pop adds: (# of registers restored) * (4 bytes) to sp to deallocate space on the stack
 - sp = sp + (# registers restored * 4)
- Remember: {reg list} must be the same in both the push and the corresponding pop

Local Variables are Part of Each Stack Frame

 Local variables are on the stack below the lowest numbered saved (pushed) register

```
#define BFSZ 4
int main(void)
{
  char buf[BFSZ]; // BFSZ bytes
...
```






4

Stack Frame (Arm Arch32 Procedure Call Standards)


Stack Frame Requirements

- Minimal frame: at function entry push {fp, lr}
- The top two entries in a stack frame are always (1) saved Ir, (2) saved fp
- sp points at top element in the stack (lowest byte address)
- fp points at the 1r copy stored in the current stack frame
- Stack frames MUST ALWAYS BE aligned to 8-byte addresses
 - So, this must always be true: sp % 8 == 0



minimal frame above
Always save at least fp and Ir
and set fp at saved Ir

```
int main(void)
{
    a();
    /* other code */
    return EXIT_SUCCESS;
}
int a(void)
{
    int x;
    int y;
    /* other code */
    return 0;
}
```






```
deallocate stack space 
SP = SP + "space"
shrinks "up"
```


Note slide has builds

FP_OFF: Distance from FP to SP Used to set FP at push and SP before pop



Function Prologue
always at top of function
saves regs and sets fp

Function Epilogue always at bottom of function restores regs including the sp



Means Caution, odd number of regs!

If odd number pushed, make sure frame
is 8-byte aligned (later)
this must always be true: sp % 8 == 0


FP_OFF = (#regs - 1)*4 // -1 is lr offset from sp
Where # regs = #preserved + lr + fp

IMPORTANT: FP_OFF has two uses:


- 1. Where to set fp after prologue push (remember sp position)
- 2. Restore sp (deallocate locals) right before epilogue pop


Function Prologue and Epilogue: Minimum Stack Frame


- Each function has only one Prologue at the top of the function body and only one Epilogue at the bottom of the function body
- When you want to exit the function, set the return value in r0, and then branch (or fall through) to the epilogue
- Function entry (Function Prologue):
 - 1. save preserved registers
 - 2. set the fp to point at saved Ir
 - allocate space for locals (subtracts from sp)
- Function return (Function **Epilogue**):
 - deallocate space for locals (adds to sp)
 - 2. restores preserved registers
 - 3. return to caller



Function Prologue: Allocating the Stack Frame








Function was just called this how the stack looks
The orange blocks are part of the caller's stack frame

Function saves Ir, fp using a push and only those preserved registers it wants to use on the stack Do not push r12 or r13


Function moves the fp to point at the saved Ir as required by the Aarch32 spec

Allocate Space for Local Variables


Function Epilogue: Deallocating the Stack Frame



Use fp as a pointer to find local variables on the stack



Move SP back to where it was after the push in the prologue. So, the pop works properly (this also deallocates the local variables)



At function exit (in the function epilogue) the function uses pop to restore the registers to the values they had at function entry

Review Return Value and Passing Parameters to Functions

(Four parameters or less)

Register	Function Call Use		
r0	1 st parameter		
r1	2 nd parameter		
r2	3 rd parameter		
r3	4 th parameter		

Register	Function Return Value Use
r0	8, 16 or 32-bit result, 32-bit address or least-significant half of a 64-bit result
r1	most-significant half of a 64-bit result

• Where r0, r1, r2, r3 are arm registers, the function declaration is (first four arguments):

- Each parameter and return value is limited to data that can fit in 4 bytes or less
- You receive up to the first four parameters in these four registers
- You copy up to the first four parameters into these four registers before calling a function
- For parameter values using more than 4 bytes, a pointer to the parameter is passed (we will cover this later)
- You MUST ALWAYS assume that the called function will alter the contents of all four registers: r0-r3
- Observation: When a function calls another function, the called function has the right to overwrite the first 4 parameters that were passed to it by the calling function

Accessing argy from Assembly (stderr version)


```
.extern printf
                                                                                % ./cipher -e -b in/B00K
    .extern stderr
                                                                              argv[0] = ./cipher
    .section .rodata
.Lstr: .string "argv[%d] = %s\n"
                                                                              argv[1] = -e
    .text
                                                                              argv[2] = -b
                  // main(r0=argc, r1=argv)
    .qlobal main
                                                                              argv[3] = in/B00K
           main, %function
    type
                                             Function Prologue
           FP OFF,
                       20
    •equ
                                             always at top of function
main:
           {r4-r7, fp, lr}
   push
                                             saves regs and sets fp
           fp, sp, FP_OFF
   add
           r4, =stderr // get the address of stderr
   ldr
           r4, [r4]
                           // get the contents of stderr
                                                                                                          in/book
   ldr
           r5, =.Lstr
                           // get the address of .Lstr
   ldr
           r6, 0
                           // set indx = 0:
   mov
                                                                                                              -b
           r7, r1
                           // save argv
                                                                    Registers
   mov
                                                                                            argv[]
.Lloop:
   // fprintf(stderr, "argv[%d] = %s\n", indx, argv[indx])
                                                                                                              - e
                                                               r3
           r3, [r7]
                          // argv[indx]
   ldr
           r3, 0
                          // check argv[indx]==NULL
   cmp
                                                               r2
                                                                                                          ./cipher
                          // if so done
           Ldone
   bea
                                                                      **argv
                                                               r1
           r2, r6
   mov
                          // indx
                          // "argv[%d] = %s\n"
           r1, r5
   mov
                                                               r0
                                                                       argc
           r0, r4
                          // stderr
   mov
           fprintf
   bl
   add
           r6, r6, 1
                          // indx++
           r7, r7, 4
   add
                          // argv++
           .Lloop
                                  Function Epilogue
.Ldone:
           r0, 0
   mov
                                  always at bottom of function Branch to this to exit the function
           sp, fp, FP_OFF
   sub
                                  restores reas including the sp
           {r4-r7, fp, lr}
   pop
    hx
```

Local Variables on the Stack


- Local variables are on the stack below the lowest numbered saved register
- frame pointer is used as a **pointer** to stack variables
- fp is the base register in ldr and str instructions
- Example load buf[0] into r4

```
#define BFSZ 4
int main(void)
{
  char buf[BFSZ]; // BFSZ bytes
...
```

- FP_OFF = 12, BUFSZ = 4
- Distance from FP is buf[0] is 12 + 4 = 16 ldrb r4, [fp, -16]
- 1. Calculate how much additional space is needed by all the local variables
- After the register save push, Subtract from the sp the size of the variable in bytes (+ padding - later slides)




Function prologue with local variables



 move the sp to allocate space on the stack for local variables and outgoing parameters (later)


```
.equ FRMADD, 8
push {fp, lr}
add fp, sp, FP_OFF
ldr r3, =FRMADD // frames may be Large
sub sp, sp, r3
// your code
```

Function epilogue with local variables



```
FRMADD, 8
.equ
push
       {fp, lr}
add
       fp, sp, FP_OFF
ldr
     r3, =FRMADD
       sp, sp, r3
sub
  // your code
      sp, fp, FP_OFF
sub
      {fp, lr}
pop
       lr // func return
bx
```


- For pop to restore the registers correctly:
 - sp must point at the last saved preserved register put on the stack by the save register operation: the push



- Return the sp (using the fp) to the same address it had after the push operation sub sp, fp, FP_OFF
- this works no matter how much space was allocated in the prologue

Stack Frame Design – Local Variables

- Arrays start at a 4-byte boundary (even arrays with only 1 element)
 - Exception: double arrays [] start at an 8-byte boundary
 - struct arrays are aligned to the requirements of largest member
- Space padding (0 or 4 bytes) when necessary is added at the high address end of a variables allocated space, based on the variable's alignment and the requirements of variable below it on the stack
- Single chars (and shorts) can be grouped together in same 4-byte word (following the alignment for the short)
- After all the variables have been allocated, add padding at stack frame bottom (low memory) so the total stack frame size (including all saved registers) is a multiple of 8 when the prologue is finished




Step 1: Stack Frame Design – Local Variables

In this example we are allocating in order of variable definition, no reordering

```
int func(void)
{
    int x = 0;
    short st[2];
    char str[] = "ABCDE";
    char *ptr = &array[0];
```


Variable name	Initial Value	Size bytes	Alignment pad to next	Total Size
int x	0	4	0	4
short a[]	; ;	2*2	0	4
char str[]	"ABCDE"	6	2	8
char *ptr	&array[0]	4	0	4
PAD Added		4		4
FRMADD (locals etc)				24
Saved Register Space		4 * 4		16
Total Frame Size				40



Accessing Stack Variables The Hard Way.....

- Access data stored in the stack
 - use ldr/str instructions
- Use base register fp with offset (distance in bytes)
 addressing (either register offset or immediate offset)
- No matter where in memory the stack is located, fp always points at saved lr)
- Word offset is a way to visualize the distance from fp for calculating offset values

Variable name	offset from fp	ldr instruction
int x	-16	ldr r0, [fp, -16]
short a[]	-20	ldrsh r0, [fp, -20]
char str[]	-28	ldrb r0, [fp, -28]
char *ptr	-32	ldr r0, [fp, -32]




Step 2 Generate Distance offsets from [fp]

 Use the assembler to calculate the distance from the address contained in fp [fp, -offset]

```
.equ FP_OFF, 12
.equ X, 4+FP_OFF // X = 16
.equ A, 4+X // A = 20
```


- Assign label names for each local variable
 - Each name is .equ to be the offset from fp

Variable name	Size	Name	expression size + prev	Distance from fp
Pushed regs-1	12	FP_OFF		12
int x	4	X	4 + FP_OFF	16
short a[]	4	А	4 + X	20
char str[]	8	STR	8 + A	28
char *ptr	4	PTR	4 + STR	32
PAD Added	4	PAD	4 + PTR	36
FRMADD		FRMADD	PAD-FP_OFF	24



Step 3 Allocate Space in the Prologue


```
.global func
  .type func, %function
  .equ FP OFF,
                    12
  .equ X, 4 + FP_OFF
  equ A, 4 + X
  .equ STR, 8 + A
  .equ PTR, 4 + STR
       PAD, 4 + PTR
  .equ
       FRMADD PAD - FP OFF
  .equ
func:
  push {r4, r5, fp, lr}
  add fp, sp, FP_OFF
  ldr r3, =FRMADD //frames can be large
  sub
     sp, sp, r3 // add space for locals
  // rest of function code
 // no change to epilogue
     sp, fp, FP_OFF // deallocate locals
  sub
  pop {r4, r5, fp, lr}
  bx
     lr
  .size func, (. - func)
```



Accessing Stack using distance offsets

var	stack v	variable address into r0	stack v	variable contents into r0
V	ldr	r0, =X	ldr	r0, =X
X	sub	r0, fp, r0	ldr	r0, [fp, -r0]
2[0]	ldr	r0, =A	ldr	r0, =A
a[0]	sub	r0, fp, r0	ldrsh	r0, [fp, -r0]
a[1]	ldr	r0, =A - 2	ldr	r0, =A - 2
a[1]	sub	r0, fp, r0	ldrsh	r0, [fp, -r0]
c+n[1]	ldr	r0, =STR - 1	ldr	r0, =STR - 1
str[1]	sub	r0, fp, r0	ldrb	r0, [fp, -r0]
n+n	ldr	r0, =PTR	ldr	r0, =PTR
ptr	sub	r0, fp, r0	ldr	r0, [fp, -r0]
	ldr	r0, =PTR	ldr	r0, =PTR
*ptr	sub	r0, fp, r0	ldr	r0, [fp, -r0]
	ldr	r0, [r0]	1dr	r0, [r0]

var	write contents of r0 to stack variable			
ptr	ldr r1, =PTR			
pti	str r0, [fp, -r1]			
*ptr	ldr r1, =PTR			
	ldr r1, [fp, -r1]			
	str r0, [r1]			



Review: Loading and using Global Variables

 Tell the assembler to create and USE a literal table to obtain the address (Lvalue) of a label into a register:

```
ldr/str Rd, =Label // Rd = address
```

• Example to the right: y = x;

two step to **load** a **memory** variable

- 1. load the pointer to the memory
- 2. read (load) from *pointer

two steps **store** to a **memory** variable

- 1. load the pointer to the memory
- 2. write (store) to *pointer

```
.bss
y :
      .space 4
       .data
       .word 200
X:
       .text
      // function header
main:
     // load the address, then contents
     // using r2
     1dr r2, =x // int *r2 = &x
      1dr r2, [r2] // r2 = *r2;
     // &x was only needed once above
     // Note: r2 was a pointer then an int
     // no "type" checking in assembly!
      // store the contents of r2
     \ldr r1, =y // int *r1 = &y
      str r2, [r1] // *r1 = r2
```

Review: Global Variable access

var	global variable address into r0				contents of r0 into global variable		
х	ldr	r0, =x	ldr ldr	r0, =x r0, [r0]			r1, =x r0, [r1]
*x	ldr ldr	r0, =x r0, [r0]	ldr ldr ldr	r0, =x r0, [r0] r0, [r0]]	ldr	r1, =x r1, [r1] r0, [r1]
у	ldr	r0, =y	ldr ldr	r0, =y r0, [r0]			r1, =y r0, [r1]
stdin	ldr	r0, =stdin	ldr ldr	r0, =sto	din]	str	r1, =stdin r0, [r1] e only, not nded)
.Lstr	ldr	r0, =.Lstr	ldr ldrb	r0, =.Ls		<re< td=""><td>ad only></td></re<>	ad only>

```
.bss
y: .space 4
stdin:.space 4 // FILE *


    .data
x: .data y //x = &y

    .section .rodata
.Lstr: .string "HI\n"
```

Step 4 Initialize the Local Variables


```
int func(void)
{
    int x = 0;
    short st[2];
    char str[] = "ABCDE";
    char *ptr = &(str[0]);
```

```
r4, 0
 mov
 ldr
       r5, =X
       r4, [fp, -r5]
 str
 ldr
        r5, =STR
       r5, fp, r5 // r5 = addr of STR
 sub
       r4, =PTR
 ldr
       r5, [fp, -r4] //ptr = &(str[0])
 str
        r4, 'A'
 mov
 strb
       r4, [r5]
       r4, 'B'
 mov
 add r5, r5, 1
 strb r4, [r5]
//...
```




C Stream Functions Array/block read/write

- Read/write ops advance the file position pointer from TOF towards EOF on each I/O
 - Moves towards EOF by number of bytes read/written
- size_t fwrite(void *ptr, size_t size, size_t count, FILE *stream);
 - Writes an array *ptr of count elements of size bytes from stream
 - Updates the write file pointer forward by the number of bytes written
 - returns number of elements written
 - Treat return != count as an error
- size_t fread(void *ptr, size_t size, size_t count, FILE *stream);
 - Reads an array *ptr of count elements of size bytes from stream
 - Updates the read file pointer forward by the number of bytes read
 - · returns number of elements read.
 - Treat a return of 0 as being in EOF state
- Set element size to 1 to return bytes read/written
- EOF is NOT a character in the file, but a condition on the stream
- int feof(FILE *stream)
 - Returns non-zero at end-of-file for stream
- int ferror(FILE *stream)
 - · Returns non-zero if error for stream



24



Passing Pointers to Stack Variables

```
#include <stdio.h>
                                                                saved Ir
                                                                                fp
 #include <stdlib.h>
 #include <errno.h>
                                                                                       Data Segment
                                                                saved fp
 #define BUFSZ 4096
                                                                                       Global Variables
                                                                saved r7
 // copies input to output
                                                                                                        struct
 int
                                                                                         FILE *
                                                                                stdin:
                                                                saved r6
 main(void) {
                                                                                stdout:
                                                                                        FILE *
                                                                                                        struct
     char buf[BUFSZ];
                                                                saved r5
     size t cnt; // assign to a register only
                                                                saved r4
     // read from stdin, up to BUFSZ bytes
     // and store them in buf
     // Number of bytes read is in cnt
     while ((cnt = fread(buf, 1, BUFSZ, stdin)) > 0) {
                                                               buf[BUFSZ]
         // write cnt bytes from buf to stdout
         if (fwrite(buf, 1, cnt, stdout) != cnt) {
             return EXIT_FAILURE;
                                                      buf[0]
                                         .text
     return EXIT SUCCESS;
                                         .global main
 }
                                                 main, %function
                                                                     // stack frame below
                                         .type
                                                  BUFSZ,
                                                              4096
                                         .equ
                                                 FP OFF,
                                                                           // fp offset in main stack frame
                                         .equ
                                                              20
                                                              BUFSZ+FP OFF// buffer
                                                  BUF,
                                         .equ
                                                  PAD,
                                                              0+BUF
                                                                           // Stack frame PAD
                                         .equ
                                                              PAD-FP OFF // space for locals+passed args
                                                 FRMADD.
                                         .equ
26
```

Reading and Writing bytes using C library routines fread() and fwrite()

```
.text
.global main
.type
       main, %function
                           // stack frame below, distances from fp
                    4096
       BUFSZ,
.equ
       FP OFF,
                    20
                                // fp offset in main stack frame
.equ
                    BUFSZ+FP OFF// buffer
.equ
       BUF,
                                // Stack frame PAD
.equ
        PAD,
                    0+BUF
                    PAD-FP OFF // space for locals+passed args
       FRMADD,
.equ
```

```
// save values in preserved registers
ldr r4, =BUF // distance from fp
sub r4, fp, r4 // pointer to buffer
ldr r5, =stdin // standard input global
ldr r5, [r5]
ldr r6, =stdout // standard output global
ldr r6, [r6]
```

```
saved Ir
saved fp
saved r7
saved r6
saved r5
saved r4

BUF=
FP_OFF +
BUFSZ

buf[BUFSZ]
```

```
// fread(buffer, element size, number of elements, FILE *)
// fread(r0=buf, r1=1, r2=BUFSZ, r3=stdin)
                            // buf
        r0, r4
mov
                            // bytes
        r1, 1
mov
        r2, BUFSZ
                           // cnt (or ldr r2, =BUFSZ)
mov
       r3, r5
mov
                            // stdin
bl
        fread
                            // check return value from fread
        r0. 0
cmp
```


```
// fwrite(buffer, element size, number of elements, FILE *)
// fwrite(r0=buf, r1=1, r2=cnt, r3=stdout)
        r0, r4
                            // buf
mov
        r1, 1
                           // bytes
mov
        r2, r7
                            // cnt
mov
        r3, r6
                            // stdout
mov
bl
        fwrite
        r0, r7
                             // check return value from fwrite
cmp
```


Passing Pointers to Stack Variables

```
#define BUFSZ 4096
int main(void) {
    char buf[BUFSZ];
    size t cnt; // assign to a register only
    while ((cnt = fread(buf, 1, BUFSZ, stdin)) > 0) {
       if (fwrite(buf, 1, cnt, stdout) != cnt) {
            return EXIT FAILURE;
       }
    return EXIT SUCCESS;
    .extern fread
    .extern fwrite
    .extern stdin
    .extern stdout
    .equ EXIT_FAILURE, 1
    .text
    .qlobal main
    .type main, %function
                        4096
    •equ
            BUFSZ,
            FP OFF.
                        20
    •equ
            BUF.
                        BUFSZ + FP OFF
    . eau
                              + BUF
    •equ
            PAD,
            FRMADD.
                        PAD-FP OFF
    . eau
// see right --→
.Ldone:
            sp, fp, FP_OFF
    sub
    pop
            {r4-r7, fp, lr}
    bx
          main, (. - main)
    .size
```

```
main:
           {r4-r7, fp, lr}
    push
    add
            fp, sp, FP OFF
                                    // set frame pointer
           r3, =FRMADD
    ldr
                                    // get frame size
           sp, sp, r3
                                   // allocate space
    sub
    // save values in preserved registers
    ldr
            r4, =BUF
                                   // distance from fp
           r4, fp, r4
                                   // pointer to buffer
    sub
    ldr
            r5, =stdin
                                   // standard input global
    ldr
            r5, [r5]
                                    // standard output global
    ldr
            r6, =stdout
    ldr
            r6, [r6]
.Lloop:
       // fread(r0=buf, r1=1, r2=BUFSZ, r3=stdin)
            r0, r4
                               // buf
   mov
            r1, 1
                               // bytes
   mov
                              // cnt (or ldr r2, =BUFSZ)
           r2, BUFSZ
   mov
           r3, r5
                                // stdin
    mov
    bl
           fread
            r0, 0
    CMD
    ble
            Ldone
            r7, r0
                                // save cnt
   mov
          // fwrite(r0=buf, r1=1, r2=cnt, r3=stdout)
            r0, r4
                               // buf
    mov
   mov
            r1. 1
                               // bytes
                               // cnt
   mov
            r2, r7
            r3, r6
                               // stdout
    mov
    bl
            fwrite
                               // did we write all the bytes?
    cmp
            r0, r7
    beq
            .Lloop
   mov
            r0, EXIT_FAILURE
.Ldone:
```

Writing Functions: Receiving a Pointer Parameter - 1





Writing Function: Receiving a Pointer Parameter - 2

```
void r0, r1, r2
fillbuf(char *s, int len, char fill)
{
   char *enptr = s + len;
   while (s < enptr)
       *(s++) = fill;
}</pre>
```

Using r1 for endptr

```
saved Ir
saved fp

buf[BFSZ-1]

buf[BFSZ]

r1

buf[0]

saved Ir
saved fp
```


```
fillbuf:
   push
         {fp, lr} // stack frame
          fp, sp, FP OFF // set fp to base
   add
          r1, r1, r0
                       // copy up to r1 = bufpt + cnt
   add
          r0, r1
                       // are there any chars to fill?
   cmp
          .Ldone
                        // nope we are done
   bge
.Ldowhile:
                       // store the char in the buffer
   strb
          r2, [r0]
          r0, 1
                       // point to next char
   add
          r0, r1
                        // have we reached the end?
   cmp
          .Ldowhile
                        // if not continue to fill
   blt
.Ldone:
          sp, fp, FP_OFF // restore stack frame top
   sub
          {fp, lr} // restore registers
   pop
                        // return to caller
   bx
          lr
```

30

Passing More Than Four Arguments – At the point of Call

r0 = function(r0, r1, r2, r3, arg5, arg6, ... argn)arg1, arg2, arg3, arg4, ...

- Args > 4 are in the <u>caller's stack frame</u> at SP (argv5), an up
- Called functions have the right to change stack args just like they can change the register args!
 - Caller must assume all args including ones on the stack are changed by the caller
- Calling function prior to making the call
 - 1. Evaluate first four args: place resulting values in r0-r3
 - 2. Store Arg 5 and greater parameter values on the stack
- One arg value per slot! NO arrays across multiple slots
 - chars, shorts and ints are directly stored
 - Structs (not always), and arrays are passed via a pointer
 - Pointers passed as output parameters usually contain an address that points at the stack, BSS, data, or heap




<u>Calling Function:</u> Allocating Stack Parameter Space

At the point of a function call (and obviously at the start of the called function):

- 1. sp must point at arg5
- 2. arg5 must be at an 8-byte boundary,
 - a) padding to force arg5 alignment is placed above the last argument the called function is expecting

Approach: Extend the stack frame to include enough space for stack arguments function with the greatest arg count

- 1. Examine every function call in the body of a function
- 2. Find the function call with greatest arg count, Determines space needed for outgoing args
- 3. Add the space needed to the frame layout




Rules: At point of call

- 1. arg5 must be pointed at by sp
- 2. SP must be 8-byte aligned

Calling Function: Pass ARGS 5 and higher

.equ	FP_OFF,4		
.equ	CNT,	8 + FP_OFF // in	t cnt[2];
.equ	PAD,	4 + CNT // added	as needed
.equ	OARG9,	4 + PAD	
.equ	OARG8,	4 + OARG9	
.equ	OARG7,	4 + OARG8	
.equ	OARG6,	4 + OARG7	
.equ	OARG5,	4 + OARG6	
.equ	FRMADD	OARG5 - FP_OFF	

var		write contents	
040051	ldr	r0, =OARG5	//distance
OARG5 = r1	str	r1, [fp, -r0]	
	ldr	r2, =CNT	//distance
	sub	r2, fp, r2	// &cnt
OARG6 = &cnt			
	ldr	r0, =OARG6	//distance
	str	r2, [fp, -r0]	



Rules: At point of call

- 1. arg5 must be pointed at by sp
- 2. SP must be 8-byte aligned

Called Function: Retrieving Args From the Stack

- At function start and before the push{} the sp is at an 8-byte boundary
- Args are in the <u>caller's stack frame</u> and arg 5 always starts at fp+4
 - Additional args are higher up the stack, with one "slot" every 4-bytes
- This "algorithm" for finding args was designed to enable variable arg count functions like printf("conversion list", arg0, ... argn);

Constant	Offset	arm ldr /str statement
ARGN	(N-4)*4	ldr r4, [fp, ARGN]
ARG9	20	ldr r4, [fp, ARG9]
ARG8	16	ldr r4, [fp, ARG8]
ARG7	12	ldrb r4, [fp, ARG7]
ARG6	8	ldr r4, [fp, ARG6]
ARG5	4	ldrh r4, [fp, ARG5]

Callers Stack frame

no defined limit to number of args, keep going up stack 4 bytes at a time

Current

Stack

Frame

.equ ARG9, 20 .equ ARG8, 16 .equ ARG7, 12 .equ ARG6, 8 .equ ARG5, 4 rest of frame

PAD

arg9

arg8

??? arg6

??? arg5

Ir to caller

callers fp

Saved

Registers

Local

variables

fp+20

fp+16

fp+12

8+qt

fp+4

saved Ir

saved fp

Rule: Called functions always access stack parameters using a positive offset to the fp

SD

Determining the Passed Parameter Area on The Stack

- Find the function called by main with the largest number of parameters
- That function determines the size of the Passed Parameter allocation on the stack

```
int main(void)
{
    /* code not shown */
    a(g, h);

/* code not shown */
    sixsum(a1, a2, a3, a4, a5, a6);

/* code not shown */

b(q, w, e, r);
    /* code not shown */
}
```

largest arg count is 6 allocate space for 6 - 4 = 2 arg slots

Passing More than Four Args – Six Arg Example


- Problem: Write and call a function that receives six integers and returns the sum
- First 4 parameters are in register r0 r3 and the remaining argument are on the stack
- For this example, we will put all the locals on the stack

```
int main(void)
{
   int cnt = sixsum(1, 2, 3, 4, 5, 6);
   printf("the sum is %d\n", cnt);
   return EXIT_SUCCESS;
}
```

```
int
sixsum(int a1, int a2, int a3, int a4, int a5, int a6)
{
    return a1 + a2 + a3 + a4 + a5 + a6;
}
```

Calling Function > 4 Args - 1


```
int cnt = sixsum(1, 2, 3, 4, 5, 6);
```



Calling Function > 4 Args - 2

```
int cnt = sixsum(1, 2, 3, 4, 5, 6);
```

```
.equ
        FP OFF, 12
                       4 + FP OFF
.equ
        CNT,
                       4 + CNT
.equ
        PAD,
        OARG6,
                        4 + PAD
.equ
        OARG5,
                        4 + OARG6
.equ
        FRMADD
                        OARG5 - FP OFF
.equ
```



```
.section .rodata
   .Lpfstr: .string "the sum is %d\n"
main:
           {r4, r5, fp, lr}
    push
           fp, sp, FP OFF
    add
           r3, =FRMADD
    ldr
           sp, sp, r3
    sub
           r0, 6
    mov
           r5, =OARG6
    ldr
           r0, [fp, -r5] // arg6
    str
           r0, 5
    mov
           r5, =0ARG5
    ldr
           r0, [fp, -r5] // arg5
    str
           r3, 4
                         // arg4
    mov
           r2, 3
                         // arg3
    mov
           r1, 2
                          // arg2
   mov
           r0, 1
                          // arg1
    mov
    bl
           sixsum
    ldr
           r5, =CNT
           r0, [fp, -r5] // update cnt on stack
    str
    mov
           r1, r0
           r0, =.Lpfstr
    ldr
    bl
           printf
           r0, EXIT SUCCESS
    mov
    sub
           sp, fp, FP OFF
           {r4, r5, fp, lr}
    pop
           lr
    bx
```

Called Function > 4 Args

```
int sixsum(int a1, int a2, int a3, int a4, int a5, int a6)
   return a1 + a2 + a3 + a4 + a5 + a6;
     ARG6, 8 // offset into caller's frame
.equ
     ARG5, 4 // offset into caller's frame
.equ
                                                        saved Ir
      FP OFF, 4 // Local base
.equ
                                                        saved fp
       sixsum:
                                                        saved r5
                    {fp, lr}
             push
             add
                    fp, sp, FP OFF
                                                        saved r4
                                           Callers
             add r0, r0, r1
                                           stack
                                                          cnt
             add r0, r0, r2
                                           frame
             add r0, r0, r3
                                                          pad
             ldr r1, [fp, ARG5]
                                                         oarq6
                                                                    fp + ARG6
             add r0, r0, r1
             ldr r1, [fp, ARG6]
                                                         oarg5
                                                                    fp + ARG5
             add r0, r0, r1
                                                        saved Ir
                                           Current
             sub sp, fp, FP OFF
                                                                    fp
                                           stack
             pop {fp, lr}
                                                        saved fp
                                           frame
                                                                     sp
             bx
                     lr
```


Extra Slides

Local Variables: Stack Frame Design Practice

Example shows allocation without reordering variables to optimize space

```
short a[3];
short *ptr1;
char tmp;
char *ptr2;
char nm[] = "cse30";
```

```
.equ
    FP OFF, 4 // Local base
// NAME, SIZE + prev_name
   A, 8 + FP OFF
.equ
   PTR1, 4 + A
.equ
    TMP, 4 + PTR1
.equ
    PTR2, 4 + TMP
.equ
    NM,
            8 + PTR2
.equ
    PAD,
            4 + NM
.equ
    FRMADD PAD - FP OFF // for locals
.equ
```



When writing real code, you do not have to put all locals on the stack


- Place locals in registers if they fit, are accessed often, and
- You do not need their address (they are not an output variable in a function call)

Local Variables: Stack Frame Design Reordering

Example shows allocation with reordering variables to optimize space

```
short a[3];
short *ptr1;
char *ptr2;
char tmp;
char nm[] = "cse30";
```


```
.equ
     FP OFF, 4 // Local base
// NAME.
       SIZE + prev name
              8 + FP_OFF
.equ A,
           4 + A
    PTR1,
.equ
     PTR2,
             4 + PTR1
.equ
     TMP, size 2 + PTR2
.equ
           change 6 + TMP
     NM,
.equ
     PAD,
                 0 + NM // not needed
.equ
                 PAD - FP_OFF
     FRMADD
.equ
```



When writing real code, you do not have to put all locals on the stack

- Place locals in registers if they fit, are accessed often, and
- You do not need their address (they are not an output variable in a function call)

ARM Assembly Source File: Header and Footer



- .syntax unified
 - use the standard ARM assembly language syntax called *Unified Assembler*
 Language (UAL)
- .section .note.GNU-stack,"",%progbits
 - tells the linker to make the stack and all data segments not-executable (no instructions in those sections) – security measure
- .end
 - at the end of the source file, everything written after the .end is ignored

Function Header and Footer Assembler Directives

```
.text
                                          .global myfunc
                                                                         // make myfunc global for linking
    function entry point
                                 Function
                                           type
                                                   myfunc, %function // define myfunc to be a function
       address of the first
                                  Header
                                                   FP OFF, 4
                                                                         // fp offset in main stack frame
                                           equ
instruction in the function
                               myfunc:
Must not be a local label
                                           // function prologue, stack frame setup
                                           // your code
 (does not start with .L)
                                           // function epiloque, stack frame teardown
                               Function
                                           size myfunc, (. - myfunc)
                                 Footer
 .global function name
    • Exports the function name to other files. Required for main function, optional for others
 .type name, %function
    • The .type directive sets the type of a symbol/label name

    %function specifies that name is a function (name is the address of the first instruction)

 equ FP OFF, 4

    Used for basic stack frame setup; the number 4 will change – later slides

 .size name, bytes

    The .size directive is used to set the size associated with a symbol

    Used by the linker to exclude unneeded code and/or data when creating an executable file

    It is also used by the debugger gdb

    bytes is best calculated as an expression: (period is the current address in a memory segment)

          In CSE30 required use: size name, (. - name)
```

Reference For PA8/9: C Stream Functions Opening Files

FILE *fopen(char filename[], const char mode[]);

- Opens a stream to the specified file in specified file access mode
 - returns NULL on failure always check the return value; make sure the open succeeded!
- Mode is a string that describes the actions that can be performed on the stream:
- "r'" Open for reading.

The stream is positioned at the beginning of the file. Fail if the file does not exist.

"w" Open for writing.

The stream is positioned at the beginning of the file. Create the file if it does not exist.

"a" Open for writing

The stream is positioned at the end of the file. Create the file if it does not exist. Subsequent writes to the file will always be at current end of file.

An optional "+" following "r", "w", or "a" opens the file for both reading and writing


Reference: C Stream Functions Closing Files and Usage

```
int fclose(FILE *stream);
```

- Closes the specified stream, if open for writing, then forcing output to complete (eventually)
 - returns EOF on failure (often ignored as no easy recovery other than a message)
- Usage template for fopen() and fclose()
 - 1. Open a file with fopen () always checking the return value
 - 2. do i/o keep calling stdio io routines
 - 3. close the file with fclose() when done with that I/O stream

C Stream Functions Array/block read/write

- Read/write ops advance the file position pointer from TOF towards EOF on each I/O
 - Moves towards EOF by number of bytes read/written
- size_t fwrite(void *ptr, size_t size, size_t count, FILE *stream);
 - Writes an array *ptr of count elements of size bytes from stream
 - Updates the write file pointer forward by the number of bytes written
 - returns number of elements written
 - Treat return != count as an error
- size_t fread(void *ptr, size_t size, size_t count, FILE *stream);
 - Reads an array *ptr of count elements of size bytes from stream
 - Updates the read file pointer forward by the number of bytes read
 - · returns number of elements read.
 - Treat a return of 0 as being in EOF state
- Set element size to 1 to return bytes read/written
- EOF is **NOT** a character in the file, but a condition on the stream
- int feof(FILE *stream)
 - Returns non-zero at end-of-file for stream
- int ferror(FILE *stream)
 - · Returns non-zero if error for stream



putchar/getcharSetting up and Usage

```
#include <stdio.h>
#include <stdlib.h>
int
main(void)
{
   int c;
   int count = 0;

   while ((c = getchar()) != EOF) {
      putchar(c);
      count++;
   }
   printf("Echo count: %d\n", count);
   return EXIT_SUCCESS;
}
```

```
.extern getchar
       .extern putchar
       .section .rodata
.Lfstr: .string "Echo count: %d\n"
       .text
       .equ EOF, -1
       .type main, %function
       .global main
       .equ FP OFF, 12
       .equ EXIT SUCCESS, 0
       push {r4, r5, fp, lr}
main:
       add fp, sp, FP OFF
       mov r4, 0 //r4 = count
/* while loop code will go here */
.Ldone:
       mov r1, r4 // count
       ldr
            r0, =.Lfstr
           printf
       bl
       mov r0, EXIT SUCCESS
       sub sp, fp, FP OFF
       pop {r4, r5, fp, lr}
       bx 1r
       .size main, (. - main)
```

Putchar/getchar: The while loop initialize count r4, 0 //count mov b1 getchar pre loop test with a call to getchar() if it returns EOF in r0 we are done r0, EOF cmp .Ldone bea .Lloop: echo the character read with getchar and b1 putchar then read another and increment count bl getchar #include <stdio.h> #include <stdlib.h> r4, r4, 1 add int r0, EOF cmp main(void) did getchar() return EOF if not loop bne .Lloop .Ldone: int c; int count = 0; mov r1, r4 ldr r0, =pfstr saw EOF, print count while ((c = getchar()) != EOF) { bl printf putchar(c); count++; printf("Echo count: %d\n", count); return EXIT SUCCESS;

File header and footers are not shown

printing error messages in assembly

```
.Lmsg0: .string "Read failed\n"
       ldr
               r0, =.Lmsg0
                                          // read failed print error
       bl
               errmsg
           // int errmsg(char *errormsg)
           // writes error messages to stderr
                 errmsg, %function
                                                 // define to be a function
           .type
                                                  // fp offset in stack frame
           .equ FP OFF,
   errmsg:
           push
                {fp, lr}
                                                 // stack frame register save
           add fp, sp, FP OFF
                                                 // set the frame pointer
                   r1, r0
           mov
               r0, =stderr
           ldr
                   r0, [r0]
           ldr
                 fprintf
           bl
           mov r0, EXIT FAILURE
                                                 // Set return value
               sp, fp, FP OFF
                                                 // restore stack frame top
           sub
           pop {fp, lr}
                                                  // remove frame and restore
                                                  // return to caller
           hx
                   1r
           // function footer
                                                 // set size for function
           .size errmsg, (. - errmsg)
```

main.S Source File Showing a minimum stack frame

```
.arch armv6
        file header
                         .arm
                         .fpu vfp
                         .syntax unified
                // BSS Segment (only when needed)
                         .bss
                // Data Segment (only when needed)
                         .data
                // Read-Only Data (only when needed)
                         .section .rodata
                // Text Segment - your code
                         .text
                                  main, %function
                         . type
 function header
                                                       4 is for
                         .global main
+ prologue constant
                                                       minimum
                                  FP OFF,
                         .equ
                                  EXIT SUCCESS, 0
                                                       stack frame
                          . equ
                                                       this will vary
                main:
                                  {fp, lr}
                         push
function prologue
                         add
                                  fp, sp, FP OFF
                         // main() code goes here
                                                         function return value in r0
                                  r0, EXIT SUCCESS
                         mov
                                  sp, fp, FP OFF
                         sub
                         pop
                                  {fp, lr}
  function epilogue
```